IHME

Graphing in Stata:

Tips, tricks, and code snippets

Tara Templin

UNIVERSITY of WASHINGTON

Institute for Health Metrics and Evaluation

Goals

- The purpose of this document is to:
  - Facilitate graphing in Stata by providing examples
  - Provide detail graphing style guidelines
  - Share code for producing publication-quality graphics

What this is and is not

- **What this is:** A practical guide of tips and tricks for data analysts, post-bachelor fellows, and researchers coding in Stata. It lays out a style guide for graphs. A likely scenario in which this would be useful is producing figures for a slide deck or publication.

- **What this is not:** A description of why R, Python, or Javascript would be better than Stata. This document assumes that you are in Stata's environment for better or worse.

Outline

1. ***Graphing Style***
2. Graphing tips and tricks
3. Auxiliary graphing items
4. Fonts and Colors

Graphing Style

1. Displaying numerical estimates

2. General graphing principles

Graphing Style: Numerical estimates

- For reporting numbers:
  - No decimal points on per capita values (e.g. $10 per person)
  - One decimal point on percentages (e.g. 10.5% of GDP)
  - Two decimal points on large absolute values (e.g. $10.53 billion)

Graphing Style: Graphing principles

- In general, your graphs should have:
  - Readable marker labels and axis values (size 6 or bigger).
  - Non-sideways y-axis titles
  - Non-sideways y-label values [ylab(, angle(90))]
  - No blue backgrounds: set scheme s1color
  - –line– instead of –scatter– for time series data
  - –rarea– instead of –graph bar, stack– for time series data that are not volatile

Outline

1. Graphing Style
2. ***Graphing tips and tricks***
3. Auxiliary graphing items
4. Fonts and Colors

Graphing Tips and Tricks

Tips and tricks for the following:

1. Bar charts
2. Line charts
3. Pie charts
4. Arrow Diagrams
5. Bubble Diagrams with arrows
6. Maps
7. Overlapping uncertainty intervals

Graphing tips and tricks: Bar charts

- Bar charts
  - Setting up your bar chart
  - Grouped bar plots
  - Overlapping bars
  - Stacked bar plots over age groups

Graphing tips and tricks: Bar chart set up

- Two types of bar charts: –gr bar– and –twoway bar– which require differently structured input data.
- A y-axis needs to be defined for –twoway bar–.
- Stacked bar charts can be created using –twoway bar– by appropriately ordering the bars so that the tallest is at the back.

Setting up the y-axis for twoway bar

```
gen n = .
local count = 2
foreach r in HIC UMC LMC LIC G {
        foreach p in dah prepaid oop ghes combo {
                di "region == `r' & payer == `p'"
                replace n = `count' if region == "`r'" & payer == "`p'"
                local count = `count' + 2
        }
        local count = `count' + 1
        if ("`r'" == "LIC") local count = `count' + 3
}
 gen m = n + 0.65
```

Graphing tips and tricks: Overlapping bars



Graphing tips and tricks: Overlapping bars

```
twoway (bar meanper1_ m  if payer == "combo", color(orange*0.7) horiz barwidth(1.5)) ///
(bar meanper1_ m if payer == "ghes", color(emerald*0.7) horiz barwidth(1.5)) ///
(bar meanper1_ m if payer == "oop", color(eltblue*0.7) horiz barwidth(1.5)) ///
(bar meanper1_ m if payer == "prepaid", color(navy*0.7) horiz barwidth(1.5)) ///
(bar meanper1_ m if payer == "dah", color(green*0.3) horiz barwidth(1.5)) ///
(bar meanper_ n  if payer == "combo", color(orange) horiz barwidth(1.5)) ///
(bar meanper_ n if payer == "ghes", color(emerald) horiz barwidth(1.5)) ///
(bar meanper_ n if payer == "oop", color(eltblue) horiz barwidth(1.5)) ///
(bar meanper_ n if payer == "prepaid", color(navy) horiz barwidth(1.5)) ///
(bar meanper_ n if payer == "dah", color(green*0.5) horiz barwidth(1.5)) ///
(rcap upperper_ lowerper_ n, color(black) msize(*0.5) horiz ), ///
 xlabel(-10 "-10%" -5 "-5%" 0 "0%" 5 "5%" 10 "10%", labsize(vsmall) glcolor(gs8) glwidth(vthin) glpattern(dot) angle(0)) ///
ylabel(4 "HIC" 8 "UMC" 12 "LMC" 16 "LIC" 24 "Global", noticks angle(0) labsize(tiny)) ///
xtitle("Percent Change in Health Spending" , size(small)) ytitle("") graphregion(fcolor(white)) legend(col(1) size(tiny)  ///
lab(6 "Total Health Spending")  lab(7 "Government Health" "Spending") lab(8 "Out-of-pocket" "Health Spending") ///
lab(9 "Prepaid Private" "Insurance") lab(10 "Development Assistance" "for Health")  order(6 7 8 9 10) symxsize(1.5) ///
position(3)  region(lcolor(white))) xline(0, lcolor(black) lwidth(thin)) aspectratio(0.8)
```

Graphing tips and tricks: Grouped bars



Graphing tips and tricks: Grouped bars

```
twoway (bar percent_ n if payer == "dah" & location_name != "HIC", color(green*0.7) msize(small) barwidth(0.9))   ///
 (bar percent_ n if payer == "oop", color(eltblue) msize(small) barwidth(0.9))   ///
 (bar percent_ n if payer == "prepaid", color(navy) msize(small) barwidth(0.9))   ///
 (bar percent_ n if payer == "ghes", color(emerald) msize(small) barwidth(0.9)),   ///
ylabel(0 "0%" 4 "4%" 8 "8%" 12 "12%" 16 "16%", labsize(vsmall) glcolor(gs12) angle(0)) ///
xlabel(`HIC_place' "High-income" `UMC_place' "Upper-middle" `LMC_place' "Lower-middle" ///
`LIC_place' "Low-income", noticks labsize(vsmall)) ///
ytitle("Health Spending per GDP (%)" "", size(small) margin(3 0 0 0)) xtitle("") ///
legend(col(1) lab(1 "DAH")  lab(2 "OOP") lab(3 "PPP") lab(4 "GHES") order(1 2 3 4) ///
size(small) symxsize(1.5) position(3) region(lcolor(white))) ///
yline(0, lcolor(black) lwidth(thin)) aspectratio(.8) scheme(s1color)
```

Graphing tips and tricks: Over categories



Graphing tips and tricks: Over categories

```
gr bar (sum) per_1 per_2 per_3 per_4 per5 per_6 per_7 per_8 per_9 per_10 ///
if Age !=99 & year == 2010, over(Age, gap(0) label( labsize(*0.5) angle(45)) ///
relabel(1 "0 - 1" 2 "1 - 4" 3 "5 - 9" 4 "10 - 14" 5 "15 - 19" 6 "20 - 24" 7 "25 - 29“ ///
8 "30 - 34" 9 "35 - 39" 10 "40 - 44" 11 "45 - 49" 12 "50 - 54" 13 "55 - 59" 14 ///
"60 - 64" 15 "65 - 69" 16 "70 - 74" 17 "75 - 79" 18 "80+")) stack name(stacked, replace)  ///
ysize(3) ylabel(0(20)100, labsize(*0.8) glcolor(gs12) glwidth(vthin) glpattern(blank) angle(90)) ///
ytitle("Percentage of total DALYs" " ",  size(*0.7)) graphregion(c(white)) ///
bar(1, c("`cancer'")) bar(2, c("`cardio'")) bar(3, c("`resp'")) bar(4, c("`cirrhosis'") ) bar(5, c("`digestive'")) ///
bar(6, c("`neuro'") ) bar(7, c("`mental'")) bar(8, c("`diabetes'")) bar(9, c("`musculo'")) bar(10, c("`other'")) ///
legend(row(2) lab(10 "Other non-communicable" "diseases") lab(9 "Musculoskeletal" "disorders") ///
lab(8 "Diabetes, urogenital," "blood, and endocrine" "diseases") lab(7 "Mental and" "behavioral disorders") ///
lab(6 "Neurological disorders") lab(5 "Digestive diseases" "(except cirrhosis)") lab(4 "Cirrhosis of the liver") ///
lab(3 "Chronic respiratory" "disease") lab(2 "Cardiovascular and" "circirculatory disease") ///
lab(1 "Neoplasms (Cancer)") order(1 2 3 4 5 6 7 8 9 10) size(small) region(lcolor(white))) ///
title("All low- and middle-income countries", size(*0.7))
```

Graphing tips and tricks: Line charts

Domestic health spending per capita



Excluding South Africa

Graphing tips and tricks: Line charts

- Line graphs: build a line graph over a bunch of categories

```
twoway (line combo_per_cap year if location_name == "Burkina Faso", lc(red*0.7)) ///
(line combo_per_cap year if  location_name == "Democratic Republic of the Congo", lc(red)) ///
(line combo_per_cap year if location_name == "Ethiopia", lc(eltblue)) ///
(line combo_per_cap year if location_name == "Ghana", lc(navy)) ///
(line combo_per_cap year if location_name == "India", lc(orange)) ///
(line combo_per_cap year if location_name == "Kenya", lc(gs8))  ///
(line combo_per_cap year if location_name == "Nigeria", lc(blue))  ///
(line combo_per_cap year if location_name == "Senegal", lc(black)) ///
(line combo_per_cap year if location_name == "Tanzania", lc(maroon)) ///
(line combo_per_cap year if location_name == "Zambia", lc(green)), ///
legend(off) ytitle("", size(small) margin(3 0 0 0))  ylabel(, angle(0)) ///
graphregion(fcolor(white)) legend(off) xlab(1995 “1995" 2010 "2010" 2025 "2025" 2040 "2040")
```

Graphing tips and tricks: Pie charts

- Pie charts: How to put the percentage on a pie slice



Graphing tips and tricks: Pie charts

```
foreach hfa in ch mh {
        graph pie ghes_`hfa' dah_`hfa' investment_`hfa', name(`hfa'_pie, replace)  ///
        pie(1, color(emerald)) pie(2, c(eltblue)) pie(3, c(red)) graphregion(c(white)) legend(col(1) pos(3) ///
        lab(1 "GHES") lab(2 "DAH") lab(3 "Gap") order(1 2 3) size(small) symxsize(1.5) region(lcolor(white))) ///
        title("", size(*0.6)) plotregion(fcolor(white) lcolor(white)) graphregion(fcolor(white) lcolor(white))  ///
        plabel(1 percent, size(smallmed) format(%12.0f)) plabel(2 percent, size(smallmed) format(%12.0f)) ///
        plabel(3 percent, size(smallmed) format(%12.0f))
}
grc1leg ch_pie mh_pie
```

Graphing tips and tricks: Arrow diagrams

- Arrow diagrams: Ranking lists and connecting them with arrows or lines



Graphing tips and tricks: Arrow diagrams

```
tw pcarrow rank_DALYS_n_pos x_daly rank_DAH_n_pos x_dah, msymbol(i) mcolor(none) lcolor(midgreen) || ///
scatter rank_DAH_n_pos  x_dah, mlabel(label_dah) mlabp(9) mlabcolor(black) msymbol(square) mcolor(dkgreen) msize(small) || ///
scatter rank_DALYS_n_pos x_daly, mlabel(label_daly) mlabp(3) mlabcolor(black) msymbol(square) mcolor(dkgreen) msize(small)  ///
text(1 1.15 "Ranking by" "`thing_1_title'", place(nw) size(small) just(right)) ///
text(1 1.45 "Ranking by" "`thing_2_title'", place(ne) size(small) just(left)) ///
xsize(5) ysize(7) graphregion(fcolor(white)) xscale(off) yscale(off) ylabel(none) ///
xsca(r(0.7 1.9)) ysca(r(2 -30)) legend(off) ///
xsca(r(0.7 1.9)) ysca(r(2 -30))
```

Graphing tips and tricks: Bubble diagrams

- Bubble diagrams connected with arrows



Bubble diagrams: weights

```
levelsof gbd_analytical_superregion_name, local(gbd_sr)
local sr_count = 1
foreach sr in `gbd_sr' {
        foreach year in 1995 2013 2040 {
                sum combo_per_cap_`year' if gbd_analytical_superregion_name == "`sr'"
                local pop_weight_`sr_count'_`year' = sqrt(`r(mean)') / 7
                di in red `pop_weight_`sr_count'_`year''
                di in red "pop_weight_`sr_count'_`year'"
        }
        local sr_count = `sr_count' + 1
}
```

Bubble diagrams: graph

```
twoway (scatter hf_transition_percent_1995 sds_score_1995 if gbd_analytical_superregion_name == "Central Europe, Eastern Europe, and Central Asia", mc(eltblue) msize(*`pop_weight_1_1995')) ///
(scatter hf_transition_percent_2013 sds_score_2013 if gbd_analytical_superregion_name == "Central Europe, Eastern Europe, and Central Asia", mc(eltblue) msize(*`pop_weight_1_2013')) ///
(scatter hf_transition_percent_2040 sds_score_2040 if gbd_analytical_superregion_name == "Central Europe, Eastern Europe, and Central Asia", mc(eltblue) msize(*`pop_weight_1_2040')) ///
(scatter hf_transition_percent_1995 sds_score_1995 if gbd_analytical_superregion_name == "Southeast Asia, East Asia, and Oceania", mc(red) msize(*`pop_weight_6_1995')) ///
(scatter hf_transition_percent_2013 sds_score_2013 if gbd_analytical_superregion_name == "Southeast Asia, East Asia, and Oceania", mc(red) msize(*`pop_weight_6_2013')) ///
(scatter hf_transition_percent_2040 sds_score_2040 if gbd_analytical_superregion_name == "Southeast Asia, East Asia, and Oceania", mc(red) msize(*`pop_weight_6_2040')) ///
(scatter hf_transition_percent_1995 sds_score_1995 if gbd_analytical_superregion_name == "High-income", mc(navy) msize(*`pop_weight_2_1995')) ///
(scatter hf_transition_percent_2013 sds_score_2013 if gbd_analytical_superregion_name == "High-income", mc(navy) msize(*`pop_weight_2_2013')) ///
(scatter hf_transition_percent_2040 sds_score_2040 if gbd_analytical_superregion_name == "High-income", mc(navy) msize(*`pop_weight_2_2040')) ///
(scatter hf_transition_percent_1995 sds_score_1995 if gbd_analytical_superregion_name == "Latin America and Caribbean", mc(midgreen) msize(*`pop_weight_3_1995')) ///
(scatter hf_transition_percent_2013 sds_score_2013 if gbd_analytical_superregion_name == "Latin America and Caribbean", mc(midgreen) msize(*`pop_weight_3_2013')) ///
(scatter hf_transition_percent_2040 sds_score_2040 if gbd_analytical_superregion_name == "Latin America and Caribbean", mc(midgreen) msize(*`pop_weight_3_2040')) ///
(scatter hf_transition_percent_1995 sds_score_1995 if gbd_analytical_superregion_name == "North Africa and Middle East", mc(orange*0.8) msize(*`pop_weight_4_1995')) ///
(scatter hf_transition_percent_2013 sds_score_2013 if gbd_analytical_superregion_name == "North Africa and Middle East", mc(orange*0.8) msize(*`pop_weight_4_2013')) ///
(scatter hf_transition_percent_2040 sds_score_2040 if gbd_analytical_superregion_name == "North Africa and Middle East", mc(orange*0.8) msize(*`pop_weight_4_2040')) ///
(scatter hf_transition_percent_1995 sds_score_1995 if gbd_analytical_superregion_name == "South Asia", mc(dkgreen) msize(*`pop_weight_5_1995')) ///
(scatter hf_transition_percent_2013 sds_score_2013 if gbd_analytical_superregion_name == "South Asia", mc(dkgreen) msize(*`pop_weight_5_2013')) ///
(scatter hf_transition_percent_2040 sds_score_2040 if gbd_analytical_superregion_name == "South Asia", mc(dkgreen) msize(*`pop_weight_5_2040')) ///
(scatter hf_transition_percent_1995 sds_score_1995 if gbd_analytical_superregion_name == "Sub-Saharan Africa", mc(mint*0.8) msize(*`pop_weight_7_1995')) ///
(scatter hf_transition_percent_2013 sds_score_2013 if gbd_analytical_superregion_name == "Sub-Saharan Africa", mc(mint*0.8) msize(*`pop_weight_7_2013')) ///
(scatter hf_transition_percent_2040 sds_score_2040 if gbd_analytical_superregion_name == "Sub-Saharan Africa", mc(mint*0.8) msize(*`pop_weight_7_2040')) ///
(pcarrow hf_transition_percent_1995 sds_score_1995 hf_transition_percent_2013 sds_score_2013, mc(black) lc(black) lwidth(medthick) msize(medium)) ///
(pcarrow hf_transition_percent_2013 sds_score_2013 hf_transition_percent_2040 sds_score_2040, mc(black) lc(black) lwidth(medthick) msize(medium)), ///
ylabel(0(10)100, glcolor(gs12) glwidth(vthin) glpattern(dot) angle(0)) xlabel(0(20)100, glcolor(gs12) glwidth(vthin) glpattern(dot) angle(0))  ytitle("Domestic health expenditure that is govermental or private insurance", size(*0.7)) ///
xtitle("Socio-demographic Status", size(*0.7))  graphregion(fcolor(white))  legend(off) yscale(r(0))  caption("Total health expenditure excludes DAH; The size of the dot is scaled to total health expenditure per capita; 2010 PPP", size(*0.7))  ///
xlabel(, labsize(*0.7)) ylabel(, labsize(*0.7)) name(hf_transition_change, replace) title("Health Financing Transition by GBD Region: 1995 to 2040", size(medsmall)) aspectratio(0.8)
```

Graphing tips and tricks: Maps

- Mapping requires:
  - Reading in .shp files
  - Appropriately binning your data
  - Producing a map

Graphing tips and tricks: Maps



Graphing tips and tricks: Reading in .shp files

```
rename iso3 iso_a3

cd "J:\Project\IRH\NCD\DATA\Maps" // Change directory -- this is where the dta files are for making the maps

** shp2dta using Somalia_Fixed, data(worlddata2) coor(worldcoord1) genid(id)  // Note -- have to download the .shp files, use shp2dta and have them in the directory.

merge m:m iso_a3 using "worlddata2.dta", keepusing(mapcolor7 mapcolor8 mapcolor9 mapcolor13 id)

replace id = 146 if iso_a3 == "SOM"

drop if id == 146 & iso_a3 == "-99"
```

Graphing tips and tricks: Binning data

```
** Save percentiles of your data
    sum `thing' if year == 2040, d
    local per5 : di %3.1f `r(p5)'
    local per10 : di %3.1f `r(p10)'
    local per25 : di %3.1f `r(p25)'
    local per50 : di %3.1f `r(p50)'
    local per75 : di %3.1f `r(p75)'
    local per90 : di %3.1f `r(p90)'
    local per99 : di %3.1f `r(p99)'
    local max : di %3.1f `r(max)' + 100
```

Graphing tips and tricks: Map legends

```
spmap `thing' using worldcoord1.dta if id!=7 & year == 2040, id(id) name(`thing', replace) ///
fcolor(RdYlGn) ocolor(black ..) osize(vvthin ..) clmethod(custom) ///
clbreaks(-100, `per5', `per10', `per25', `per50', `per75', `per90', `max') ///
legend(on) legend(region(lcolor(white) fcolor(white))) ///
legend(title("Government health spending as a share of total") ///
col(1) lab(2 "0% to `per5'%") lab(3 "`per5'% to `per10'%") lab(4 "`per10'% to `per25'%") ///
lab(5 "`per25'% to `per50'%") lab(6 "`per50'% to `per75'%") lab(7 "`per75'% to `per90'%") ///
lab(8 "More than `per90'%") order(8 7 6 5 4 3 2)) ///
title("Percent of health spending financed by the government: 2040", size(medsmall))
```

Graphing tips and tricks: Overlapping UIs

How to create the illusion of overlapping UIs



Graphing tips and tricks: Overlapping UIs

```
twoway (rarea oop_gdp_upper oop_gdp_lower year if iso3=="CHN", col(red) fintensity(30)) ///
    (rarea ghes_gdp_upper ghes_gdp_lower year if iso3=="CHN", col(blue) fintensity(30)) ///
    (rarea ghes_gdp_lower oop_gdp_upper year if iso3=="CHN" & ///
    ghes_gdp_lower < oop_gdp_upper & year > 2012, col(purple) fintensity(30)) ///
    (line oop_gdp year if iso3 == "CHN", lc(red) lwidth(medthick)) ///
    (line ghes_gdp year if iso3 == "CHN", lc(navy) lwidth(medthick)), ///
    ylabel(#4, labsize(*0.6) glcolor(gs12) glwidth(vthin) glpattern(blank) angle(0)) ///
    ytitle("% of GDP", size(*0.7)) xtitle("Year", size(*0.7)) graphregion(c(white)) ///
    legend(off) xlabel(1995(5)2040, labs(small)) ///
    name(GHES, replace) yscale(range(0 6)) title("")
```

Outline

1. Graphing Style
2. Graphing tips and tricks
3. ***Auxiliary graphing items***
4. Fonts and Colors

Auxiliary graphing items

- Combining plots
- Parallelizing graphs on the cluster

Auxiliary graphing items: Combining Plots

- –graph combine– has some useful options for ensuring the graphs' aspect ratios remain acceptable:
  - ysize(10)
  - xsize(10)
  - altshrink

- The user written –grc1leg– is useful if there is one common legend for all graphs

Auxiliary graphing items: Parallelizing graphs

- I/O error when attempting to write parallelized PDFs on the computing cluster
- Way around it using –pdfappend–
- Dealing with country names can be burdensome, see example

Auxiliary graphing items: Parallelizing graphs

```
levelsof name if iso3 == "`c'", local(temp1)
local y = subinstr(`temp1',`"""',"",.)
local country_title: list clean temp1
di `temp1'
di `"`country_title'"'
pdfstart using `"$path/check_countries_`y'.pdf"'
<graph commands>
graph combine ghes oop prepaid dah combo total, row(2) title("Preliminary Estimates: `y'", size(*0.8))
pdfappend
pdffinish
```

Auxiliary graphing items: Parallelizing graphs



Name	Date modified	Type	Size
check_countries_Afghanistan	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Albania	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Algeria	5/9/2016 6:31 PM	Adobe Acrobat D...	11 KB
check_countries_Andorra	5/9/2016 6:30 PM	Adobe Acrobat D...	10 KB
check_countries_Angola	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Antigua and Barbuda	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Argentina	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Armenia	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Australia	5/9/2016 6:30 PM	Adobe Acrobat D...	10 KB
check_countries_Austria	5/9/2016 6:30 PM	Adobe Acrobat D...	10 KB
check_countries_Azerbaijan	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Bahrain	5/9/2016 6:30 PM	Adobe Acrobat D...	10 KB
check_countries_Bangladesh	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Barbados	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Belarus	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Belgium	5/9/2016 6:30 PM	Adobe Acrobat D...	10 KB
check_countries_Belize	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB
check_countries_Benin	5/9/2016 6:30 PM	Adobe Acrobat D...	11 KB

Outline

1. Graphing Style
2. Graphing tips and tricks
3. Auxiliary graphing items
4. ***Fonts and Colors***

Fonts and Colors

- Fonts
  - Installing custom fonts and exporting them to PDF/EPS

- Colors
  - FGH Health Focus Areas
  - GBD

Fonts and colors

- There may be a custom font required for graphing (eg Haarlemmer)
- To install it:
  - Open the file and copy the entirety of its contents
  - Navigate to Control Panel -> Fonts
  - Paste the contents into the Fonts folder

- To export graphs with the font:

```
graph export "file.eps", fontface("Haarlemmer MT") orientation(landscape) replace
```

Fonts and Colors

Health focus area	STATA color used
HIV/AIDS	emerald
Maternal/Child	eltblue
Malaria	purple*0.9
Tuberculosis	purple*0.2
Noncommunicable	orange_red*0.5
Other	midgreen*0.8

GBD Level 1	STATA color used
Communicable	red
NCDs	blue
Injuries	green

IHME

Graphing in Stata:

Tips, tricks, and code snippets

UNIVERSITY of WASHINGTON

Institute for Health Metrics and Evaluation